ΠΑΡΑΡΤΗΜΑ Β (2ου ΜΑΘΗΜΑΤΟΣ): ΠΑΡΑΔΕΙΓΜΑΤΑ WINBUGS

1 EXAMPLE 1: BIRTHWEIGHT & ESTRIOL LEVEL

```
{
#       definition of likelihood function
#
        for (i in 1:n) {
                birth[i]~dnorm( mu[i], tau ); # random component
                mu[i]<-a.star+b*(estriol[i]-mean(estriol[]));             # systematic component
                                                                          #      & link function
        }
        #       prior distributions
        #
        a.star~dnorm( 0, 1.0E-04 ); # normal prior for a
        b~dnorm( 0, 1.0E-04 ); # normal prior for b
        tau~dgamma( 1.0E-04 , 1.0E-04 ); # gamma prior for precision
        s2<-1/tau;
        a<-a.star-b*mean(estriol[]);
}
list(a.star=0.0, b=0.0, tau=1.0) # initial values

# data
list(n=31)
estriol[] birth[]
7       25
9       25
9       25
12      27
14      27
16      27
16      24
14      30
16      30
16      31
17      30
19      31
21      30
24      28
15      32
16      32
17      32
25      32
27      34
15      34
15      34
15      35
16      35
19      34
18      35
17      36
18      37
20      38
22      40
25      39
24      43
END
```


a.star
b
tau
s2
a
mu[i]
estriol[i]
birth[i]
for(i IN 1 : n)

2 Example 2: BEETLES DATASET

```
model beetles;
{
 for (i in 1:N) {
   r[i] ~ dbin(p[i], n[i]);
   logit(p[i])   <- alpha.star + beta*(x[i]-mean(x[]));
   r.hat[i] <- p[i]*n[i];  # fitted values
 }
 alpha.star ~ dnorm(0.0, 1.0E-3);
 beta      ~ dnorm(0.0, 1.0E-3);
 alpha     <- alpha.star - beta*mean(x[]);
 odds.ratio <- exp( beta )
}
```

```
list(x = c(1.6907, 1.7242, 1.7552, 1.7842, 1.8113, 1.8369, 1.8610, 1.8839)     n = c(59, 60, 62, 56, 63, 59, 62, 60)     r = c(6, 13, 18, 28, 52, 53, 61, 60), N=8)  #data
list(alpha.star=0, beta=0) #initial values
```



3 EXAMPLE 3: ΧΡΗΣΗ ΝΕΑΣ ΠΙΘΑΝΟΦΑΝΕΙΑΣ (ΓΕΝΙΚΕΥΜΕΝΗ POISSON)

```
model
{
C<-10000
for (i in 1:9) {
        zeros[i]<-0
        zeros[i]~dpois( lambda[i] )
        lambda[i]<- C - loglike[i]
        loglike[i] <- log(zeta)+(y[i]-1)* log(zeta+omega*y[i])-(zeta+omega*y[i])-logfact(y[i])
        }
        zeta~dgamma(0.001, 0.001)
        omega~dbeta(1,1)

        mean<-zeta/(1-omega)
        var<-zeta/pow(1-omega,3)
        DI<-1/((1-omega)*(1-omega))
}
```

DATA

```
list( y=c(24, 13, 7,  18, 2, 10, 3, 9, 16)  )
```

INITS

```
list( zeta=1, omega=0.5 )
```

4 EXAMPLE 4: ΠΡΟΣΟΜΟΙΩΣΗ ΣΤΟ ΠΑΡΑΣΚΗΝΙΟ ΤΟΥ ΠΑΡΑΔΕΙΓΜΑΤΟΣ ESTRIOL

4.1 APXEIO script.odc

```
display('log')
check('c:/myfiles/courses/BUGS_course/01_presentations/Bugs_files/lec2/ex1_estriol_back/model.odc')
data('c:/myfiles/courses/BUGS_course/01_presentations/Bugs_files/lec2/ex1_estriol_back/data.odc')
data('c:/myfiles/courses/BUGS_course/01_presentations/Bugs_files/lec2/ex1_estriol_back/data2.odc')
compile(1)
inits(1,'c:/myfiles/courses/BUGS_course/01_presentations/Bugs_files/lec2/ex1_estriol_back/inits.odc')

update(1000)
set(a)
set(b)
update(1000)
trace(*)
update(1000)
stats(*)
history(*)
density(*)
autoC(*)
quantiles(*)
dic.stats()
coda(*,output)
save('seedsLog')
```

4.2 APXEIO model.odc

```
{
#       definition of likelihood function
#
        for (i in 1:n) {
                birth[i]~dnorm( mu[i], tau ); # random component
                mu[i]<-a.star+b*(estriol[i]-mean(estriol[]));      # systematic component
                                                                   #      & link function
        }
#       prior distributions
#
        a.star~dnorm( 0, 1.0E-04 );                                # normal prior for a
        b~dnorm( 0, 1.0E-04 );                                     # normal prior for b
        tau~dgamma( 1.0E-04 , 1.0E-04 );                           # gamma prior for precision
        s2<-1/tau;
        a<-a.star-b*mean(estriol[]);
}
```

4.3 APXEIO data.odc

```
estriol[] birth[]
7       25
9       25
9       25
12      27
14      27
16      27
16      24
14      30
16      30
16      31
17      30
19      31
21      30
24      28
15      32
16      32
17      32
25      32
27      34
15      34
15      34
15      35
16      35
19      34
18      35
17      36
18      37
20      38
22      40
25      39
24      43
END
```

4.4 APXEIO data2.odc

```
list(n=31)
```

4.5 APXEIO inits.odc

```
list(a.star=0.0, b=0.0, tau=1.0) # initial values
```